/*

Florian Keusch 433
Proposal title: Impact of Response Scale Direction on Survey Responses to Factual/Behavioral Questions

HYPOTHESES

Stated-Hyp 1: The direction of the scale affects survey responses.

	Test-Hyp 1: Participants in the high to low scale direction condition report a different average mental health than participants in the low to high scale direction condition.

	Test-Hyp 2: Participants in the high to low scale direction condition report a different average of healthy behaviors than participants in the low to high scale direction condition.

Stated-Hyp 2: I hypothesize a larger order effect for vertical presentation of the frequency scale than for horizontal presentation, i.e. a higher endorsement of the first scale points when the scale is presented vertically compared with horizontally.

	Test-Hyp 3: There will be a significant scale direction condition x scale alignment condition interaction effect on mental health.

	Test-Hyp 4: There will be a significant scale direction condition x scale alignment condition interaction effect on healthy behaviors.

Stated-Hyp 3: I hypothesize that frequency scales using vague quantifiers will show a bigger scale presentation order effect and that precise verbal labels will reduce the influence of scale presentation order.

	Test-Hyp 5: There will be a significant scale direction condition x label vagueness condition (vague vs precise) interaction effect on mental health.

	Test-Hyp 6: There will be a significant scale direction condition x label vagueness condition (vague vs vague and precise) interaction effect on mental health.

	Test-Hyp 7: There will be a significant scale direction condition x label vagueness condition (precise vs vague and precise) interaction effect on mental health.

	Test-Hyp 8: There will be a significant scale direction condition x label vagueness condition (vague vs precise) interaction effect on healthy behaviors.

	Test-Hyp 9: There will be a significant scale direction condition x label vagueness condition (vague vs vague and precise) interaction effect on healthy behaviors.

	Test-Hyp 10: There will be a significant scale direction condition x label vagueness condition (precise vs vague and precise) interaction effect on healthy behaviors.
	
********************************************************************************

NOTES:

The proposal mentions many other analyses in the analysis plan section. However, 
here we focus on the most common metric (the mean).

"Analysis plan
The key independent variable for the analysis is the direction of the response scales. To analyze the main effect of scale direction, I plan to start with an examination of response distributions and variances for each individual question, and variance-covariance structure among the two sets of related survey items (mental health and health behavior) under different scale directions. To carry the research further, I propose to study the factor patterns and the predictive power of factors generated. In addition, I will use the answers to the open ended questions and the GfK profile variables to investigate validity. To explore the interaction effect of scale direction and the other scale features, I will repeat the same sets of analyses crossing scale direction and the other two features to identify scale features that induce larger scale order effect or reduce scale order effect. The findings will help me make recommendations on how to best design and present frequency scales for survey practitioners.
To distinguish mechanisms that are at work for the scale direction effect, I will repeat the analyses by respondent characteristics to examine any subgroup differences or interactions between scale direction and respondent characteristics. I will focus on characteristics of respondents indicative of either respondent ability or respondent motivation to assess satisficing theory. I speculate that respondent ability and motivation affects the size of scale direction effect. In addition, satisficing respondents are less likely to be affected by scale alignment and vagueness of scale verbal labels whereas vagueness of scale verbal labels may affect the use of anchor and heuristics. In addition, the scale alignment may affect the use of the “near-meansrelated” heuristics. By combining the scale features and respondent characteristics, I hope to identify or differentiate respondents for whom different mechanisms are at work. Multi-way ANCOVA models will be fit to analyze the effect of the scale direction in combination with the two other experimental factors (scale alignment and vagueness of labels) controlling for age, gender, and education.
Finally, I will examine response times to these survey items and use the results to help understand the mechanism of scale direction effect. Again, presumably satisficing respondents should take less time to answer survey questions regardless of experimental conditions whereas the use of anchor-and-adjustment heuristics takes time."
	
*/

clear all
use "TESS3_150_Keusch.dta"

********************************************************************************
* RECODING

* Scale Direction Condition (based on table on page 7 in questionnaire doc)
	cap drop scaleDirectionCondition
	tab xtess150, mis
	tab xtess150, nolabel
	gen scaleDirectionCondition = 0
	replace scaleDirectionCondition = 1 if xtess150 < 7
	label define scaleDirectionCondition_lab 0 "Low to High" 1 "High to Low" 
	label values scaleDirectionCondition scaleDirectionCondition_lab  	
	tab xtess150 scaleDirectionCondition, mis
	tab dov_direct scaleDirectionCondition, mis

* Scale Alignment Condition (based on table on page 7 in questionnaire doc)
	cap drop scaleAlignmentCondition
	tab xtess150, mis
	tab xtess150, nolabel
	gen scaleAlignmentCondition = 0
	replace scaleAlignmentCondition = 1 if xtess150 < 4 | (xtess150 > 6 & xtess150 < 10)
	label define scaleAlignmentCondition_lab 0 "Horizontal" 1 "Vertical" 
	label values scaleAlignmentCondition scaleAlignmentCondition_lab  	
	tab xtess150 scaleAlignmentCondition, mis
	tab dov_align scaleAlignmentCondition, mis

* Label Vagueness Condition (based on table on page 7 in questionnaire doc)
	cap drop labelVaguenessCondition
	tab xtess150, mis
	tab xtess150, nolabel
	gen labelVaguenessCondition = 0
	replace labelVaguenessCondition = 1 if xtess150 == 2 | xtess150 == 5 | xtess150 == 8 | xtess150 == 11
	replace labelVaguenessCondition = 2 if xtess150 == 3 | xtess150 == 6 | xtess150 == 9 | xtess150 == 12
	label define labelVaguenessCondition_lab 0 "Vague" 1 "Precise" 2 "Vague and Precise"
	label values labelVaguenessCondition labelVaguenessCondition_lab
	tab xtess150 labelVaguenessCondition, mis
	tab dov_label labelVaguenessCondition, mis	

* DV: Mental Health (higher values = more healthy)
	cap drop dvMental
	tab xtess150 q1a_1, mis
	tab xtess150 q1a_2, mis
	tab xtess150 q1a_3, mis
	tab xtess150 q1a_4, mis
	tab xtess150 q1a_5, mis
	tab xtess150 q1a_6, mis
	recode q1a_1 q1a_2 q1a_3 q1a_4 q1a_5 q1a_6 (-1 = .)
	recode q1b_1 q1b_2 q1b_3 q1b_4 q1b_5 q1b_6 (-1 = .)
	recode q1c_1 q1c_2 q1c_3 q1c_4 q1c_5 q1c_6 (-1 = .)
	recode q1d_1 q1d_2 q1d_3 q1d_4 q1d_5 q1d_6 (-1 = .)
	recode q1e_1 q1e_2 q1e_3 q1e_4 q1e_5 q1e_6 (-1 = .)
	recode q1f_1 q1f_2 q1f_3 q1f_4 q1f_5 q1f_6 (-1 = .)
	gen dvMental = .
	replace dvMental = q1a_1 + q1b_1 + q1c_1 + q1d_1 + q1e_1 + q1f_1 if xtess150 == 4 | xtess150 == 10
	replace dvMental = q1a_2 + q1b_2 + q1c_2 + q1d_2 + q1e_2 + q1f_2 if xtess150 == 5 | xtess150 == 11
	replace dvMental = q1a_3 + q1b_3 + q1c_3 + q1d_3 + q1e_3 + q1f_3 if xtess150 == 6 | xtess150 == 12
	replace dvMental = q1a_4 + q1b_4 + q1c_4 + q1d_4 + q1e_4 + q1f_4 if xtess150 == 1 | xtess150 == 7
	replace dvMental = q1a_5 + q1b_5 + q1c_5 + q1d_5 + q1e_5 + q1f_5 if xtess150 == 2 | xtess150 == 8
	replace dvMental = q1a_6 + q1b_6 + q1c_6 + q1d_6 + q1e_6 + q1f_6 if xtess150 == 3 | xtess150 == 9
	tab dvMental, mis

* DV: Healthy Behavior (higher values = more healthy)
	cap drop dvBehavior
	tab xtess150 q2a_1, mis
	tab xtess150 q2a_2, mis
	tab xtess150 q2a_3, mis
	tab xtess150 q2a_4, mis
	tab xtess150 q2a_5, mis
	tab xtess150 q2a_6, mis
	recode q2a_1 q2a_2 q2a_3 q2a_4 q2a_5 q2a_6 (-1 = .)
	recode q2b_1 q2b_2 q2b_3 q2b_4 q2b_5 q2b_6 (-1 = .)
	recode q2c_1 q2c_2 q2c_3 q2c_4 q2c_5 q2c_6 (-1 = .)
	recode q2d_1 q2d_2 q2d_3 q2d_4 q2d_5 q2d_6 (-1 = .)
	gen dvBehavior = .
	replace dvBehavior = q2a_1 + q2b_1 + q2c_1 + q2d_1 if xtess150 == 4 | xtess150 == 10
	replace dvBehavior = q2a_2 + q2b_2 + q2c_2 + q2d_2 if xtess150 == 5 | xtess150 == 11
	replace dvBehavior = q2a_3 + q2b_3 + q2c_3 + q2d_3 if xtess150 == 6 | xtess150 == 12
	replace dvBehavior = q2a_4 + q2b_4 + q2c_4 + q2d_4 if xtess150 == 1 | xtess150 == 7
	replace dvBehavior = q2a_5 + q2b_5 + q2c_5 + q2d_5 if xtess150 == 2 | xtess150 == 8
	replace dvBehavior = q2a_6 + q2b_6 + q2c_6 + q2d_6 if xtess150 == 3 | xtess150 == 9
	tab dvBehavior, mis

********************************************************************************

* ANALYSIS

* Test-Hyp 1: Participants in the high to low scale direction condition report a different average mental health than participants in the low to high scale direction condition.
	reg dvMental i.scaleDirectionCondition
	// no support for H1 (p = .857)
	tess 1.scaleDirectionCondition, init(Keusch433) bonf(2)

	
* Test-Hyp 2: Participants in the high to low scale direction condition report a different average of healthy behaviors than participants in the low to high scale direction condition.
	reg dvBehavior i.scaleDirectionCondition
	// no support for H2 (p = .418)
	tess 1.scaleDirectionCondition, bonf(2)
	
	
* Test-Hyp 3: There will be a significant scale direction condition x scale alignment condition interaction effect on mental health.
	reg dvMental i.scaleDirectionCondition##i.scaleAlignmentCondition
	// no support for H3 (p = .375)
	tess 1.scaleDirectionCondition#1.scaleAlignmentCondition, bonf(2)

	
* Test-Hyp 4: There will be a significant scale direction condition x scale alignment condition interaction effect on healthy behaviors.
	reg dvBehavior i.scaleDirectionCondition##i.scaleAlignmentCondition
	// no support for H4 (p = .747)
	tess 1.scaleDirectionCondition#1.scaleAlignmentCondition, bonf(2)
	
	
* Test-Hyp 5: There will be a significant scale direction condition x label vagueness condition (vague vs precise) interaction effect on mental health.
	reg dvMental i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 2
	// no support for H5 (p = .514)
	tess 1.scaleDirectionCondition#1.labelVaguenessCondition, bonf(2)
	
	
* Test-Hyp 6: There will be a significant scale direction condition x label vagueness condition (vague vs vague and precise) interaction effect on mental health.
	reg dvMental i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 1
	// no support for H6 (p = .617)
	tess 1.scaleDirectionCondition#2.labelVaguenessCondition, bonf(2)
	
	
* Test-Hyp 7: There will be a significant scale direction condition x label vagueness condition (precise vs vague and precise) interaction effect on mental health.
	reg dvMental i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 0
	// no support for H7 (p = .868)
	tess 1.scaleDirectionCondition#2.labelVaguenessCondition, bonf(2)
		

* Test-Hyp 8: There will be a significant scale direction condition x label vagueness condition (vague vs precise) interaction effect on healthy behaviors.
	reg dvBehavior i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 2
	// no support for H8 (p = .515)
	tess 1.scaleDirectionCondition#1.labelVaguenessCondition, bonf(2)
	
	
* Test-Hyp 9: There will be a significant scale direction condition x label vagueness condition (vague vs vague and precise) interaction effect on healthy behaviors.
	reg dvBehavior i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 1
	// no support for H9 (p = .137)
	tess 1.scaleDirectionCondition#2.labelVaguenessCondition, bonf(2)
	
	
* Test-Hyp 10: There will be a significant scale direction condition x label vagueness condition (precise vs vague and precise) interaction effect on healthy behaviors.
	reg dvBehavior i.scaleDirectionCondition##i.labelVaguenessCondition if labelVaguenessCondition != 0
	// no support for H10 (p = .453)
	tess 1.scaleDirectionCondition#2.labelVaguenessCondition, bonf(2)
